
1 Introduction

This report documents an improved implementation of the Harvest Rate Model (HRM), a

simple mathematical model used to provide management advice on Klamath River fall-run

chinook salmon. An additional goal of this report is to set forth explicitly the equations

that form the Harvest Rate Model. It appears that there has been no previous written

documentation of this model.

The original implementation of the HRM was a spreadsheet in Lotus 1–2–31; here, that

implementation is termed the Lotus Harvest Rate Model or LHRM. The new implementation

is termed the Quick Harvest Rate Model or QHRM. The underlying mathematical model,

without regard to implementation, is simply termed the Harvest Rate Model or HRM.

The HRM is used to estimate, in three segments of the fishery, the maximum harvest

rates that are consistent with the escapement required under management regulations.

At the same time, the harvest rates must be in accord with harvest-sharing agreements

among harvesting segments. Finding the appropriate harvest rates with the LHRM requires

extensive trial and error. The QHRM, in contrast, computes the harvest rates without

manual intervention, given the desired constraints and relationships among harvests. The

QHRM is currently implemented as a Fortran program.

2 The Harvest Rate Model

In this section, the HRM is explained in words and equations; the progression from LHRM

to the QHRM is detailed in §3. The model uses the variables listed in Table 1; depending

on the context, many of them may be used with zero, one, or two subscripts. When two

subscripts are used, the first indicates segment of the fishery, and the second indicates

age. A variable with a single subscript is either age-independent (if a rate) or is a sum

across ages (if a number of fish). A variable with no subscript is a sum across all ages and

segments. As is a frequent practice in salmon biology, instantaneous rates are not used;

the rates involved are simple proportions.

2.1 Known and Unknown Quantities

The quantities in Table 1 are of three types: constraints, unknowns, and biological param-

eters. The constraints are the required escapement in natural areas E′, the proportion of

1Use of tradenames does not imply endorsement by the National Marine Fisheries Service.

Table 1. List of Variables

Variable Description

a subscript denoting age, a ∈ {3,4,5}
s subscript denoting segment of fishery, s ∈ {f , o, t, r}
f previous-fall segment
o ocean segment
t tribal segment
r river-recreational segment
ρ tribal plus river-recreational segments
C number of fish contacted (caught) by fishery
D dropoff rate (as proportion of fish contacting gear)
E spawning escapement (number of fish)
E′ spawning escapement in natural areas (number of fish)
F harvest rate (proportion); definition varies by fishery
H number of fish harvested (landed) by fishery
I impact (number of fish killed by fishery)
L proportion of fish of legal size
Ma proportion mature at age a
N number of fish in ocean prior to ocean fishery
N′ number of fish in river prior to river fisheries
ν proportion of spawning that occurs in natural areas
Pt proportion of total harvest taken by tribal fisheries
Pr proportion of non-tribal harvest taken by river-recreational fishery
R shaker (catch-and-release) mortality rate (proportion)
S shaker or dropoff deaths (number of fish)
V vulnerability to gear or “contact rate” (proportion)

harvest allocated to the tribes Pt , and the proportion of the non-tribal harvest allocated to

the river-recreational segment Pr . The values of these constraints are set by the manage-

ment plan and harvest-sharing agreements. The unknowns are the ocean harvest rate Fo,
the tribal harvest Ht , and the river recreational harvest Hr . (It is explained in §3 that, due

to the constraints, there is really only one unknown, Fo, and that for any Fo there are unique

corresponding values of Ht and Hr .) The remaining quantities in Table 1 are biological

parameters that are presumed known. In practice, estimates of them are available.

The equations below assume that the three harvest rates are known. This provides a

logical presentation and is similar to the assumption used in the LHRM. In §3, the method

of solving for the actual unknowns is given.

2.2 Ocean Impacts

The HRM sequentially models the ocean, tribal, and river-recreational segments of the

fishery. Computations for the ocean fishery begin with estimates of the starting ocean

population size at age Na; these estimates are typically available from cohort reconstruc-

tion analyses. The first quantity modeled is the number of fish Co,a of age a contacted by

the ocean fishery. This is the product of the number in the ocean (adjusted for harvest in

the previous fall), the vulnerability, and the ocean harvest rate:

Co,a = (Na −Hf,a)Vo,a Fo .(1)

Simple summation is used for the total number of ocean contacts:

Co =
5∑

a=3

Co,a .(2)

(Two issues of terminology require explanation. We have followed the LHRM in referring

to the quantity Fo as the ocean harvest rate; however, inspection of equation (1) reveals

that it might more correctly be called the ocean encounter rate. It is certainly not a harvest

rate in any usual sense of the word. Nonetheless, we were hesitant to change the name of

a quantity so central to this model, and have thus retained the term “harvest rate” in all

places that it is used in the LHRM. The second issue concerns what we call vulnerability V ,

and what the LHRM calls the “contact rate.” We consider the LHRM terminology misleading.

The term vulnerability is widely used in marine fisheries models for the concept, if not the

precise quantity, expressed here by V .)

The ocean harvest rate Fo is assumed age-independent, but the product Vo,a Fo is in essence

the age-specific harvest rate. From the number of fish contacted, the age-specific ocean

harvest deaths Ho,a and shaker (catch-and-release) deaths So,a are modeled:

Ho,a = Co,a La +Hf,a(3)

and

So,a = (Co,a −Ho,a)Ra .(4)

Equation (4) is the formulation used by the LHRM. However, it is flawed in that the fall

harvest was added to the regular-season harvest to make up the ocean harvest in equation

(3), but this total ocean harvest is subtracted when computing shaker mortality in equation

(4). This flaw should make little difference in the computations, because fall harvests are

typically very small. However, it is logically inconsistent and should be corrected. In our

only departure from the LHRM, we therefore used the corrected equation:

So,a = (Co,a − Co,a La)Ra .(5)

Ocean impacts at age are defined as the sum of ocean harvest and shaker mortalities:

Io,a = Ho,a + So,a .(6)

The modeled totals of these quantities are obtained by summing across ages:

Ho =
5∑

a=3

Ho,a ,(7)

So =
5∑

a=3

So,a ,(8)

Io =
5∑

a=3

Io,a .(9)

2.3 Tribal Impacts

After the ocean fishery has been modeled, the number of fish remaining in the ocean at

age is considered to be Na − Io,a. The resulting age-specific river run N′a is that number

times the proportion mature, i.e.,

N′a = (Na − Io,a)Ma .(10)

In the HRM, the two river fisheries (tribal and recreational) are treated as independent,

rather than competing, sources of mortality. The modeled tribal harvest at age is computed

from the tribal harvest rate, the tribal dropoff rate, the number of fish in the river, and the

vulnerability of those fish:

Ht,a = Ft(1−Dt)N′aVρ,a .(11)

The number of deaths attributed to the tribal dropoff rate is then modeled:

St,a = Ft Dt N′aVρ,a .(12)

(As with the ocean fishery, terminology needs clarification. As can be seen from equation

(12), the term dropoff rate refers to fish that are killed by the gear but not landed. The

term harvest rate is used here (and for the river-recreational segment) in an idiosyncratic

sense; for these two segments, FV is in truth an impact rate that includes the effects of

dropoffs. However, we follow the LHRM’s usage in this respect.)

Finally, tribal impact at age is defined as the sum of harvest and dropoff deaths:

It,a = Ht,a + St,a .(13)

As with the ocean fishery, the model considers total tribal harvest, dropoff deaths, and

impacts to be simple sums:

Ht =
5∑

a=3

Ht,a ,

St =
5∑

a=3

St,a ,

It =
5∑

a=3

It,a .

2.4 River-Recreational Impacts

Computations for this segment of the fishery are exactly analogous to those for the tribal

segment. Harvest is modeled as

Hr,a = Fr (1−Dr)N′aVρ,a .(14)

The number of deaths attributed to the river-recreational dropoff rate is then

Sr,a = Fr Dr N′aVρ,a .(15)

6



The impact of this segment is modeled as the sum of harvest and dropoff deaths:

Ir ,a = Hr,a + Sr,a .(16)

Again, the model uses simple summation for totals:

Hr =
5∑

a=3

Hr,a ,

Sr =
5∑

a=3

Sr,a ,

Ir =
5∑

a=3

Ir ,a, .

2.5 Escapement

Spawning escapement is computed as the river run less the impacts of the river fisheries:

E =
5∑

a=3

N′a − It − Ir .(17)

An age-averaged estimate of the proportion ν of spawners in natural areas is used to model

the total escapement in natural areas:

E′ = νE .(18)

Thus, from the set of harvest rates {Fo, Ft, Fr}, the LHRM can project whether the escape-

ment goal is met. Projected values of harvest ratios Pt and Pr can also be computed and

compared to the harvest-sharing agreements. However, the LHRM provides no way, except

trial and error, of finding the harvest rates that meet that goal and those agreements.

3 The Quick Harvest Rate Model (QHRM)

The model structure embodied in equations (1) through (18) was formalized by the authors

from the LHRM spreadsheet. The same relationships were then incorporated into the Quick

Harvest Rate Model, which overcomes some limitations of the LHRM.

3.1 Formulation

Given the model structure outlined above, can the appropriate harvest rates be found other

than by trial and error? Examination of the equations reveals that the answer is yes: Be-

cause of the harvest-sharing agreements, just one tribal harvest and one river-recreational

harvest are consistent with a given ocean harvest Ho and corresponding ocean harvest

rate Fo. Thus, given an ocean harvest rate, the corresponding tribal and river-recreational

impacts can be used to project the resultant spawning in natural areas. The problem then

becomes one of finding the ocean harvest rate that corresponds to the desired escapement

in natural areas. We now explain how this is done.

The development given here assumes that the proportion spawning in natural areas ν is

constant across ages. This assumption is used in the LHRM because age-specific estimates

of ν are not available. However, the QHRM computer program can accept age-specific

values of ν and perform the corresponding computations correctly.

Ocean harvest is related to Fo through equations (1) and (3). Using the sharing agreements

(defined by Pt and Pr in Table 1) and the equations given in §2, one can derive expressions

for the tribal and river-recreational impacts in terms of the ocean harvest as follows. To

meet the sharing agreements, the tribal harvest in terms of the ocean harvest must be

Ht = Pt Ho
(1− Pt)(1− Pr ) .(19)

The corresponding tribal harvest rate will be

Ft = Ht
(1−Dt)

∑5
a=3N

′
aVρ,a

.(20)

In analogy to equation (12), the number of dropoff deaths from the tribal fishery will be

St = FtDt
5∑

a=3

N′aVρ,a.(21)

To meet the sharing agreements, the river-recreational harvest in terms of the ocean har-

vest must be

Hr = Pr Ho
(1− Pr ) .(22)

It follows that the corresponding river-recreational harvest rate will be

Fr = Hr
(1−Dr)

∑5
a=3N

′
aVρ,a

.(23)

The projected number of dropoff deaths due to the river-recreational fishery will be

Sr = FrDr
5∑

a=3

N′aVρ,a.(24)

The total projected impact Iρ of the river fisheries will be the sum of the individual impacts:

Iρ ≡ It+ Ir = Ht+ St + Hr+ Sr .(25)

Escapement is defined as the number of fish returning to spawn N′ less the river impacts

Iρ . We show below that bothN′ and Iρ are linearly related to the number of ocean contacts,

which is a linear function of the ocean harvest rate Fo. These linear relationships provide

a straightforward solution for the ocean harvest rate providing the desired escapement in

natural areas.

Equations (3) and (19–25) imply that

Iρ = KHo = K
5∑

a=3

(Hf,a + Co,aLa) ,(26)

where the proportionality constant K is defined as

K = Pt
(1− Pt)(1− Pr )(1−Dt) +

Pr
(1− Pr )(1−Dr) .(27)

Equations (3), (5), (6), and (10) imply that

N′a =
(
Na −Hf,a − Co,aLa − Co,a(1− La)Ra

)
Ma .(28)

Inserting these relationships into equations (17) and (18) shows that escapement to natural

areas declines linearly with the ocean harvest rate:

E′ = α− βFo ,(29)

where the proportionality constants α and β are defined:

α = ν
5∑

a=3

(Na −Hf,a)Ma −Hf,aK ,(30)

β = ν
5∑

a=3

(
Ma(La + Ra − RaLa)+ LaK

)(
Na −Hf,a

)
Vo,a .(31)

Given a natural escapement goal Ĕ′, the HRM thus concludes that the ocean harvest rate

F̆o that will allow that goal to be met is

F̆o = α− Ĕ′
β

.(32)

Note that α and β are known quantities in that that they depend entirely on the constraints

and biological parameters considered known in the model (user inputs). Thus the trial-

and-error approach of the LHRM is unnecessary. Given Pt , Pr , the biological parameters,

and an escapement goal, the appropriate harvest rates, along with the resulting harvests

and impacts, can be found directly.

3.2 Use of the QHRM Computer Program

The authors have implemented the QHRM as a Fortran program with a very basic user

interface, so that the Fortran code is relatively independent of the computer system or

operating environment. The program always reads its input from a file named QHRM.INP.

A sample input file is given in Appendix A.1. To protect the user’s data, the input file

should be made with another name and copied to QHRM.INP before running the QHRM.

The program is then started by typing the executable name, either QHRM1 for DOS or QHRM2

for OS/2.

Except for a few screen messages that indicate the program is running, all output is written

to the file QHRM.OUT. A sample output file corresponding to the sample input file is given

in Appendix A.2. To prevent the output’s being overwritten, the output file should be

renamed before the QHRM is run again. The input and output files are plain ASCII, so they

can be printed or edited easily.

The authors will gladly provide machine-readable copies of the source code and executable

files. The source code is also given in Appendix B. The program was written for Watcom

Fortran 77 and includes a few extensions specific to that compiler; with a few minor mod-

ifications, it should work with any modern Fortran compiler.

A Appendix: Sample Input and Output Files

A.1 Listing of QHRM.INP (sample input file)

’Klamath Quick Harvest Rate Model - Test Data (H95FF_1)’ ## Title, < 70 chars.
3.5d4 ## Desired "natural" escapement
0.50d0 ## Desired tribal share (proportion of total)
0.12d0 ## Desired river rec share (proportion of non-tribal)
0.88d0 1.0d0 1.0d0 ## ## Ocean vulnerability (contact rate) at age
0.8d0 1.0d0 1.0d0 ## Proportion legal at age in ocean
0.25d0 0.0d0 0.0d0 ## Shaker mortality rate at age
0.337d0 0.936d0 1.0d0 ## Proportion mature at age
1.345d5 3.76d4 1.6d3 ## Starting stock size at age
0.0d0 1.75d2 5.0d0 ## Previous fall harvest at age
0.59d0 1.0d0 1.0d0 ## River vulnerability (contact rate) at age
6.2d-1 6.2d-1 6.2d-1 ## Proportion spawning in natural areas at age
8.0d-2 2.0d-2 ## Dropoff rates - tribal, river rec
---------EOF----------

A.2 Listing of QHRM.OUT (sample output file)

QHRM: Klamath Quick Harvest Rate Model (ver. 1.00) 25 Jan 1998 at 14:18
By Michael Prager and Michael Mohr, NMFS Tiburon.
Based on a spreadsheet by USFWS, Arcata.

Run title: Klamath Quick Harvest Rate Model - Test Data (H95FF_1)

USER INPUT PARAMETERS
--------------------------------------------------------------------------------

Ocean Prop Shaker Tribal Rec Prop
vulnerab legal Prop mort dropoff dropoff natural

Age rate size mature rate rate rate spawners
--------------------------------------------------------------------------------

3 0.880 0.800 0.337 0.250 0.080 0.020 0.620
4 1.000 1.000 0.936 0.000 0.080 0.020 0.620
5 1.000 1.000 1.000 0.000 0.080 0.020 0.620

--------------------------------------------------------------------------------

PROJECTED OCEAN IMPACTS (* = input data)
--------------------------------------------------------------------------------

* Begin * Take Ocean Proj Proj Proj Proj
stock in prev harvest ocean ocean shaker ocean

Age size fall rate contact harvest deaths impact
--------------------------------------------------------------------------------

3 134500. 0. 0.099 11754. 9403. 588. 9991.
4 37600. 175. 0.099 3717. 3892. 0. 3892.
5 1600. 5. 0.099 158. 163. 0. 163.

Sum 173700. 180. . 15629. 13458. 588. 14046.
--------------------------------------------------------------------------------

PROJECTED RIVER IMPACTS (* = input data)
--------------------------------------------------------------------------------

Remaining Adult * River River Proj Proj Proj
ocean river vulnerab impact river river river

Age stock run rate rate harvest dropoffs impact
--------------------------------------------------------------------------------

3 124509. 41960. 0.59 0.19 7343. 586. 7930.
4 33708. 31551. 1.00 0.32 9359. 747. 10106.
5 1437. 1437. 1.00 0.32 426. 34. 460.

Sum 159654. 74947. . . 17128. 1367. 18496.
--------------------------------------------------------------------------------

FINAL HARVEST AND ESCAPEMENT PROJECTIONS (ADULT FISH)
--------------------------------------------------------------------------------
Total harvest 30586.
Ocean harvest 13458.
River harvest 17128.
Tribal harvest 15293. (50.0% of total harvest)
Non-tribal harvest 15293.
River recreational harvest 1835. (12.0% of non-tribal harvest)
Total spawning escapement 56452.
Spawning escapement in nat areas 35000.
--------------------------------------------------------------------------------

For documentation of this computer program, see NMFS Southwest Fisheries
Science Center Administrative Report T-97-01, available free from Librarian,
SW Fisheries Science Center, 3150 Paradise Drive, Tiburon, CA 94920.

B Appendix: Listing of Fortran Code for QHRM

B.1 Listing of QHRM.FOR (main Fortran code)

C =====================================================================
PROGRAM QHRM

C
C Klamath quick harvest rate model F77 M. Prager November 1996
C (revised by M. Mohr January 1997)
C
C This version finds the ocean harvest rate that will meet
C the escapement goal, based on tribal and river recreational
C fisheries that must be a fixed proportion of the total harvest.
C
C Based on a spreadsheet by USFWS, Arcata.
C Translated into Fortran and features added by
C Michael H. Prager, SW Fisheries Science Center,
C National Marine Fisheries Service, Tiburon, California
C (415) 435-3149, ext. 221 mikep@aurora.tib.nmfs.gov
C ---------------------------------------------------------------------
C

implicit none
real version
double precision Fo,FoCALC
parameter (version=1.0)
include ’qhrm.fi’

C
write (*,400)

C ...Read input file
call readdata

C
C ...Enough fish even with no ocean harvest?

Fo = 0d0
call evaluate(Fo)
if (escnattot .lt. escgoal) then

write (*,410)
stop

end if
C
C ...Ocean harvest rate

Fo = FoCALC()
C
C ...Projected quantities

call evaluate(Fo)
C
C ...Write out results

call wrtout(version)
write (*,420)

C
400 format(1x,’*** QHRM running...’)
410 format(1x,’*** QHRM ERROR: Escapement goal is too high to meet,’,

& ’ even with zero harvest.’)
C
420 format(1x,’*** QHRM done.’)
C

end
C =====================================================================
C

DOUBLE PRECISION FUNCTION FoCALC()

C implicit none
include ’qhrm.fi’
double precision a,a1,a2,b,b1,b2,b3,c,k
integer i

C
C ...Sharing agreements

k = ptribe / ((1d0-ptribe)*(1d0-prr)*(1d0-dtr))
k = k + ( prr / ((1d0-prr)*(1d0-drr)) )

C
C ...Initialize sum variables

a = 0d0
b = 0d0

C ...Alpha and beta coefficients.

do i = aa,az
C

c = n0(i) - hfall(i)
C

a1 = pmat(i) * c
a2 = hfall(i) * k
a = a + ( pnat(i) * (a1-a2) )

C
b1 = (1d0-plegal(i)) * mshak(i)
b2 = (plegal(i) + b1) * pmat(i)
b3 = plegal(i) * k
b = b + ( pnat(i) * (b2+b3) * c * vo(i) )

C
end do

C
C ...Desired ocean harvest rate

FoCALC = (a - escgoal) / b
C

return
END

C =====================================================================
C

SUBROUTINE EVALUATE(Fo)

C implicit none
include ’qhrm.fi’
double precision Fo,ret,sum,trate,rrrate
integer i

C
C ...Copy the ocean harvest rate to each age

do i = aa,az
orate(i) = Fo

end do
C
C ...Initialize Sum variables

hocntot = 0d0
hrivtot = 0d0
hfalltot = 0d0
impocntot = 0d0
remtot = 0d0
rruntot = 0d0
n0tot = 0d0
potcontot = 0d0
ncontot = 0d0
dshaktot = 0d0

C Evaluate ocean harvest & impacts:
C

do i = aa, az ! For all ages
potcon(i) = (n0(i)-hfall(i))*vo(i) ! Potential contacts
ncon(i) = orate(i) * potcon(i) ! Actual contacts
hocn(i) = plegal(i) * ncon(i) + hfall(i) ! Ocean harvest
dshak(i) = (ncon(i)-hocn(i)) * mshak(i) ! Shaker deaths
impocn(i) = hocn(i) + dshak(i) ! Ocean impacts
rem(i) = n0(i) - impocn(i) ! Remaining fish
rrun(i) = rem(i) * pmat(i) ! River run

C ...Sum quantities:
hfalltot = hfalltot + hfall(i) ! Fall landings
impocntot = impocntot + impocn(i) ! Ocean impacts
remtot = remtot + rem(i) ! Fish remaining after ocean hvst
rruntot = rruntot + rrun(i) ! River run
potcontot = potcontot + potcon(i) ! Potential ocean contacts
n0tot = n0tot + n0(i) ! Fish at start of season
ncontot = ncontot + ncon(i) ! Actual contacts
dshaktot = dshaktot + dshak(i) ! Shaker deaths
hocntot = hocntot + hocn(i) ! Ocean landings

end do
C
C Evaluate the tribal catch:
C
C ... Compute what the tribal catch should be:

htribetot = (hocntot * ptribe) / ((prr-1d0)*(ptribe-1d0))
C ... Compute the corresponding harvest rate:

ret = 1d0 - dtr ! ret = retention rate
sum = 0d0
do i = aa, az

sum = sum + rrun(i) * vr(i)
end do
trate = htribetot / (ret * sum)

C ... Break down tribal harvest by age:
do i = aa,az

htribe(i) = trate * ret * rrun(i) * vr(i)
end do

C ... Compute number of dropoffs:
tdroptot = 0d0
do i = aa,az

tdrop(i) = trate * dtr * rrun(i) * vr(i)
tdroptot = tdroptot + tdrop(i)

end do
C
C Evaluate the river recreational catch
C
C ... Compute what the river rec catch should be:
C ... PRR is the river rec catch as a proportion of non-tribal catch

hrrtot = hocntot * prr / (1d0 - prr)
C ... Compute the corresponding harvest rate:

ret = 1d0 - drr ! Retention rate
rrrate = hrrtot / (ret * sum)

C ... Break down river rec harvest by age:
do i = aa,az

hrr(i) = rrrate * ret * rrun(i) * vr(i)
end do

C ... Compute number of dropoffs:
rrdroptot = 0d0
do i = aa,az

rrdrop(i) = rrrate * drr * rrun(i) * vr(i)
rrdroptot = rrdroptot + rrdrop(i)

end do
C
C Total river impacts and escapement

escapetot = 0d0
escnattot = 0d0
imprivtot = 0d0
hrivtot = htribetot + hrrtot
do i = aa,az

impriv(i) = htribe(i) + tdrop(i) + hrr(i) + rrdrop(i)
imprivtot = imprivtot + impriv(i)
escape(i) = rrun(i) - impriv(i) ! Escapement at age
escapetot = escapetot + escape(i) ! Escapement at all ages
escnat(i) = escape(i) * pnat(i) ! " " " in natural areas
escnattot = escnattot + escnat(i) ! " " " " in natural areas
hriv(i) = htribe(i) + hrr(i) ! River harvest
pimpriv(i) = impriv(i) / rrun(i) ! River impact rate

end do
C

return
end

C =====================================================================
C

SUBROUTINE WRTOUT(version)
C

implicit none
include ’qhrm.fi’
real version
double precision temp,temp2,temp3,temp4
integer i
character*3 month(12)
integer*2 yr,mth,day,hr,min,sec,hsec

C
data month/’Jan’,’Feb’,’Mar’,’Apr’,’May’,’Jun’,’Jul’,’Aug’,
& ’Sep’,’Oct’,’Nov’,’Dec’/

C
open (unit=25, file=’QHRM.OUT’, status=’unknown’,action=’write’)

C
C ...Get date & time (Watcom style)

call getdat(yr,mth,day)
call gettim(hr,min,sec,hsec)

C ...Output heading:
write (25,300) version,day,month(mth),yr,hr,min
write (25,310) title

C
C ...Input parameters table:

write (25,340)
do i = aa,az

write (25,350) i,vo(i),plegal(i),pmat(i),mshak(i),dtr,
& drr,pnat(i)
end do
write (25,250)

C
C ...Ocean projections table:

write (25,360)
do i = aa,az

write (25,370) i,n0(i),hfall(i),orate(i),ncon(i),
+ hocn(i),dshak(i),impocn(i)
end do
write (25,380) n0tot, hfalltot,ncontot,hocntot, dshaktot,impocntot

C
C ...River projections table:

write (25,390)
do i = aa,az

temp = rrdrop(i) + tdrop(i)
write (25,400) i, rem(i),rrun(i),vr(i),pimpriv(i),

+ hriv(i),temp,impriv(i)
end do
temp = rrdroptot + tdroptot
write (25,410) remtot,rruntot,hrivtot,temp,imprivtot

C
C ... Final harvest and escapement table:

temp = htribetot + hrrtot + hocntot
temp2 = (htribetot / temp) * 100d0
temp3 = hrrtot + hocntot
temp4 = (hrrtot / temp3) * 100d0
write (25,430) temp,hocntot,hrivtot,htribetot,temp2,temp3,
& hrrtot,temp4,escapetot,escnattot

C
C ... Closing note:

write (25,440)
C
250 format (80(’-’))
300 format (’QHRM: Klamath Quick Harvest Rate Model (ver. ’,F4.2,’)’,

& t61,i2.2,1x,a3,’ ’i4,’ at ’,i2.2,’:’,i2.2)
310 format(6x,’By Michael Prager and Michael Mohr, NMFS Tiburon.’/,

& 6x,’Based on a spreadsheet by USFWS, Arcata.’,//,
& ’Run title: ’,a69)

C
C ...Table 1 follows:
340 format (//’USER INPUT PARAMETERS’,/,80(’-’),/,

& t10,’Ocean’,t22,’Prop’,
& t42,’Shaker’,t53,’Tribal’,t67,’Rec’,t77,’Prop’/,
& t7’vulnerab’,t21,’legal’,t33,’Prop’,t44,’mort’,t52,’dropoff’,
& t63,’dropoff’,t74,’natural’,/,’Age’,t11’rate’,t22,’size’,
& t31,’mature’,t44,’rate’,t55,’rate’,t66,’rate’,t73,’spawners’,
& /,80(’-’))

350 format (I3,8(5x,f6.3))
C
C ... Table 2 follows:
360 format (//,’PROJECTED OCEAN IMPACTS’,t65,’(* = input data)’,

& /,80(’-’),/,t8,’* Begin’,t20,
& ’* Take’,t33,’Ocean’,t44,’Proj’,t55,’Proj’,t66,’Proj’,
& t77,’Proj’,/,t10’stock’,
& t19,’in prev’,t31,’harvest’,t43,’ocean’,t54,’ocean’,
& t64,’shaker’,t76,’ocean’,/,’Age’,t11’size’,t22,’fall’,
& t34,’rate’,t41,’contact’,t52,’harvest’,t64,’deaths’,t75,
& ’impact’,/,80(’-’))

370 format(I3,1x,2(2x,f9.0),2x,f9.3,4(2x,f9.0))
380 format(’Sum’,1x,2(2x,f9.0),7x,’.’,3x,4(2x,f9.0),/,80(’-’))
C
C ... Table 3 follows:
390 format (//,’PROJECTED RIVER IMPACTS’,t65,’(* = input data)’,

& /,80(’-’),/,t6,’Remaining’,t21,’Adult’,
& t30,’* River’,t43,’River’,t55,’Proj’,t66,’Proj’,
& t77,’Proj’,/,t10’ocean’,t21,’river’,
& t29,’vulnerab’,t42,’impact’,t54,’river’,
& t65,’river’,t76,’river’,/,’Age’,t10’stock’,t23,’run’,
& t33,’rate’,t44,’rate’,t52,’harvest’,t62,’dropoffs’,t75,
& ’impact’,/,80(’-’))

400 format(I3,1x,2(2x,f9.0),2(1x,f9.2,1x),3(2x,f9.0))
410 format(’Sum’,1x,2(2x,f9.0),2(7x,’.’,3x),3(2x,f9.0),/,80(’-’))

C ...Table 4 follows:
430 format (//’FINAL HARVEST AND ESCAPEMENT PROJECTIONS’,

& ’ (ADULT FISH)’,/,80(’-’)/,
& ’Total harvest’,t40,f9.0,/,
& ’Ocean harvest’,t40,f9.0,/,
& ’River harvest’,t40,f9.0,/,
& ’Tribal harvest’,t40,f9.0,
& 3x,’(’,f4.1,’% of total harvest)’,/,
& ’Non-tribal harvest’,t40,f9.0,/,
& ’River recreational harvest’,t40,f9.0,
& 3x,’(’,f4.1,’% of non-tribal harvest)’,/,
& ’Total spawning escapement’,t40,f9.0,/,
& ’Spawning escapement in nat areas’,t40,f9.0,/,80(’-’))

C ...Closing note:
440 format(/’For documentation of this computer program, see’,

& ’ NMFS Southwest Fisheries’/’Science’
& ’ Center Administrative Report’,
& ’ T-97-01, available free from Librarian,’/’SW Fisheries’,
& ’ Science Center, 3150 Paradise Drive, Tiburon, CA 94920.’)

C
return
END

C =====================================================================
C

SUBROUTINE READDATA
C

implicit none
logical exists
integer i
include ’qhrm.fi’

C
C Check that the input file exists:
C

INQUIRE (FILE=’QHRM.INP’, EXIST=exists)
IF (.NOT. exists) THEN

write (*,450)
stop

endif
450 format (1x,’QHRM ERROR: The input file must be named QHRM.INP.’/

+ 1x,’A file with that name was not found in the’,
+ ’ current directory.’)

C
open (unit=20,file=’QHRM.INP’,status=’old’,action=’read’)

C
C Start reading the data
C

read (20,*) title
read (20,*) escgoal ! Desired "natural" escapement
read (20,*) ptribe ! Tribal portion of harvest
read (20,*) prr ! Rec portion of non-tribal harvest
read (20,*) (vo(i),i=aa,az) ! Ocean vulnerability rate

C Note: vulnerability is sometimes called "contact rate"
read (20,*) (plegal(i),i=aa,az) ! Ocean prop legal @ age
read (20,*) (mshak(i),i=aa,az) ! Ocean shaker rate @ age
read (20,*) (pmat(i),i=aa,az) ! Proportion mature @ age
read (20,*) (n0(i),i=aa,az) ! Starting pop size @ age
read (20,*) (hfall(i),i=aa,az) ! Fall harvest @ age
read (20,*) (vr(i),i=aa,az) ! River vulnerability @ age
read (20,*) (pnat(i),i=aa,az) ! Prop that spawn in nat areas @ age
read (20,*) dtr, drr ! Dropoff rates - tribal, riv rec
close (unit=20)

C
return
end

C END OF QHRM
C =====================================================================

17



B.2 Listing of QHRM.FI (Fortran include file)

C======================================================================
C QHRM.FI Include file for QHRM -- Prager -- November 1996
C

character*70 title
integer aa,az
parameter (aa=3,az=5)
double precision dtr,drr,escgoal,hfalltot
double precision prr,htribe(aa:az),htribetot
double precision hocntot,remtot,rruntot,imprivtot,hrivtot
double precision ptribe,n0tot,potcontot,ncontot,dshaktot
double precision impocntot,escapetot,escnattot
double precision vo(aa:az),plegal(aa:az),mshak(aa:az)
double precision pmat(aa:az),n0(aa:az)
double precision hfall(aa:az),vr(aa:az),pnat(aa:az)
double precision potcon(aa:az),hocn(aa:az),ncon(aa:az)
double precision dshak(aa:az),impocn(aa:az),impriv(aa:az)
double precision orate(aa:az),hriv(aa:az)
double precision escape(aa:az),escnat(aa:az),rem(aa:az)
double precision rrun(aa:az),pimpriv(aa:az)
double precision tdrop(aa:az),tdroptot,hrr(aa:az),hrrtot
double precision rrdrop(aa:az),rrdroptot

C
common/dpvars/prr,htribe,htribetot,hrr,hrrtot,hfalltot,
$ hocntot,remtot,rruntot,imprivtot,hrivtot,ptribe,n0tot,
$ potcontot,ncontot,dshaktot,impocntot,escapetot,escnattot,
$ vo,plegal,mshak,pmat,n0,hfall,vr,pnat,escgoal,
$ potcon,hocn,ncon,dshak,impocn,impriv,orate,hriv,
$ escape,escnat,rem,rrun,pimpriv,tdrop,tdroptot,
$ rrdrop,rrdroptot,drr,dtr

C
common/cvars/title

C =====================================================================

C Appendix: Corrections in the Second Printing

C.1 Specific Corrections

This revision of Adminstrative Report T–97–01 incorporates several corrections from the

original printing of February 3, 1997. In the original printing, a transcription error caused

the vulnerability coefficients Vρ,a to be omitted from the formulas giving the impacts of

river fisheries, both tribal and recreational. However, the V coefficients did appear cor-

rectly in the Fortran source code printed in the Appendices. This revised printing includes

the necessary corrections to equations (11), (12), (14), (15), (20), (21), (23), and (24). There

are two corresponding corrections to the text: the sentence introducing equation (11) now

mentions the vulnerability, and a sentence near the top of page 6 now states that FV (rather

than F as in the original printing) is an impact rate.

Two other minor changes were made. On page 4, the sentence introducing equation (2)

has been corrected to refer to the number of ocean contacts, rather than the ocean harvest.

Finally, in the sample data-input file reproduced in Appendix A.1, the comment on the final

line has been modified to make it clear that the value refers only to the river-recreational

fishery.

C.2 Obtaining the Revised Report

This second printing has been made electronically; pending further work on the topic,

no additional printed copies of the report have been published. Electronic copies in the

.PDF format can be obtained by anonymous ftp to dana.tib.nmfs.gov. The report can

be found in directory /pub/qhrm as file qhrm.pdf. Printed copies in the final format can

be made from the .PDF file.

